Filter by URL Value
On certain web pages you have a set of filters, but to make it easier for users, you may want to create a link to a pre-filtered page by passing the filter value on the URL.  For example, to display a certain supplier or a certain category of products, you might create a link such as:

http://localhost/Training1/Products/ShowProductsTablePage.aspx?Supplier=Exotic%20Liquids&Category=Beverages
The Supplier=Exotic Liquids and Category=Beverages are passed on the URL.

You can accomplish this easily by overriding two methods in the Products Table Control class on a Show Products Table page. This class corresponds to the table being displayed. 

1. Override the CreateWhereClause method to add the URL parameters to the where clause when the page is first displayed (is not a postback).  

2. Override the Populate Filter method to set the current value of the filter based on the URL parameter.  This is done for each of the filters displayed on the web page.  If there is no filter displayed on the page, this step is not required.

Override Create Where Clause

In this method, we first call the base CreateWhereClause to add any where clauses from generated in Section 2 for the where clauses specified on the panel wizard.

Next we check the IsPostBack flag since we only want to display the filtered values when the page is first displayed.  Subsequent changes made by the end-user on the page will not automatically add the URL parameters to the where clause.

Finally, we add the values to the Where Clause.  Note that most often the value contained in the table is an Id for the filter.  For example, the Products Table will contain the Supplier Id and the Category Id.  However, the values passed by the user may be the Display Foreign Key As (DFKA) value such as “Exotic Liquids” and not an Id such as 2.  In this case, we need to let the Where Clause know that the value passed may need to be “expanded” prior to comparison.  This is accomplished by passing True for the fourth parameter to the iAnd function indicating that foreign keys should be expanded.
Public Overrides Function CreateWhereClause() As WhereClause

    Dim wc As WhereClause = MyBase.CreateWhereClause()

    If (IsNothing(wc)) Then
        wc = New WhereClause

    End If
    ' Return where clause if this is a postback.  Only use URL parameters on first load.
    If Me.Page.IsPostBack Then Return wc

    ' Get the url value
    Dim Supplier As String = Me.Page.Request.QueryString("Supplier")

    Dim Category As String = Me.Page.Request.QueryString("Category")

    If Not (IsNothing(Supplier)) Then wc.iAND(ProductsTable.SupplierID, BaseFilter.ComparisonOperator.EqualsTo, Supplier, True)

    If Not (IsNothing(Category)) Then wc.iAND(ProductsTable.CategoryID, BaseFilter.ComparisonOperator.EqualsTo, Category, True)

    ' if you only want to filter the table when the page loads first time
    ' then you need to check for Postback and add whereclause if its not a Postback    
    ' Add additional Where Clause here    
    Return wc

End Function

Override Populate Filter

The Populate Filter methods first call the based populate filter function.  After calling the base method, the URL parameter value is retrieved and if this is not a postback invocation, the URL value is looked up in the dropdown filter and selected.
Protected Overrides Sub PopulateSupplierIDFilter(ByVal selectedValue As String, ByVal maxItems As Integer)

    MyBase.PopulateSupplierIDFilter(selectedValue, maxItems)

    Dim urlValue As String = Me.Page.Request.QueryString("Supplier")

    If Not Me.Page.IsPostBack AndAlso Not IsNothing(urlValue) Then
        Dim li As ListItem = Me.SupplierIDFilter.Items.FindByText(urlValue)

        If Not (IsNothing(li)) Then li.Selected = True
    End If
End Sub
This process is repeated for each of the filters that can be set by URL.

Protected Overrides Sub PopulateCategoryIdFilter(ByVal selectedValue As String, ByVal maxItems As Integer)

    MyBase.PopulateCategoryIdFilter(selectedValue, maxItems)

    Dim urlValue As String = Me.Page.Request.QueryString("Category")

    If Not Me.Page.IsPostBack AndAlso Not IsNothing(urlValue) Then
        Dim li As ListItem = Me.CategoryIdFilter.Items.FindByText(urlValue)

        If Not (IsNothing(li)) Then li.Selected = True
    End If
End Sub
